Pensieve header: The primary program accompanying “Over then Under Tangles”, by Dror Bar-Natan, Zsuzsanna Dancso, and Roland van der Veen.

This notebook also generates the LaTeX file SomeComputations.tex, that corresponds to the section “Some Computations” in the paper.

Mathematica and LaTeX Initialization

In[ ]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\OU"]
```

Out[ ]=
```
C:\drorbn\AcademicPensieve\Projects\OU
```

tex

```
{
\def\face{\input{figs/face.pdf_t}}
\def\human{\input{figs/human.pdf_t}}
\def\machine{\input{figs/machine.pdf_t}}

\def\cellscale{1}
\newsavebox\pdfbox
\newlength{\pdfheight}

\def\nbpdfInput#1{{%
 \savebox{\pdfbox}{\includegraphics[scale=\cellscale]{#1}}%
 \settoheight{\pdfheight}{\usebox{\pdfbox}}%
 %\uselengthunit{mm}\printlength{\pdfheight}%
 \noindent\imagetop{\ifdim\pdfheight<10mm\face\else\human\fi}\ %
 \imagetop{\usebox{\pdfbox}}%
 \vskip 2mm%
}}

\def\nbpdfOutput#1{{\noindent{\imagetop{\machine}\
\imagetop{\includegraphics[scale=\cellscale]{#1}}\vskip 2mm}}}

\def\m#1{\text{\tt #1}}
```

Section Introduction

tex

```
\subsection{Some Computations} \label{ssec:comp}
```

We believe in implementing as much as possible. Actually, we hardly believe ourselves unless we implement.

All code here is written in {\sl Mathematica}~\cite{Wolfram:Mathematica} and is available as the {\sl Mathematica} notebook {\sl SomeComputations.nb} at~\cite{Self}.

Virtual Diagrams



tex

\parpic[r]{\input{figs/VDExample.pdf_t}}
To represent a virtual tangle diagram D on the computer, we order its strands and traverse each of them in order, marking each ``O'' point, each ``U'' point, and each end of strand with the integers $1,2,3,\ldots$ in the order in which they are encountered. (See examples on the right). For each crossing x of D we form an expression $\m{X}_s[i,j]$, where s is the sign of the crossing and i and j are the markings next to O side and the U side of x, respectively. We also for an expression $\m{EOS}[k]$ for each end-of-strand marked k. We toss all this information into a container \m{VD}, and the result is our computer representation of D. Below, \m{vd1} and \m{vd2} are the results of this process for the two example tangles on the right.

pdf

In[ ]:=
```
SetAttributes[VD, Orderless]
```

pdf

In[ ]:=
```
vd1 = VD[X_{+1}[1, 4], X_{+1}[5, 2], EOS[3], EOS[6]];
vd2 = VD[X_{+1}[1, 4], X_{+1}[2, 7], X_{+1}[6, 3], EOS[5], EOS[8]];
```

tex

Sometimes in a \m{VD} we allow to label O/U/\m{EOS} points by arbitrary real numbers, for in fact, only the ordering of these points matter. The routine \m{Tidy} takes a real-ordered \m{VD} and converts it to a sequentially ordered one. Thus it brings a \m{VD} to a ``canonical form'':

pdf

In[ ]:=
```
Tidy[vd_VD] := Module[{ps = Union @@ (List @@@ vd)},
  Replace[vd, Thread[ps → Range@Length@ps], {2}]]
```

pdf

In[ ]:=
```
VD[X_{+1}[0.9, 4.2], X_{+1}[5, e], EOS[π], EOS[60]] // Tidy
```

pdf

Out[ ]=
```
VD[EOS[4], EOS[6], X_1[1, 2], X_1[3, 5]]
```

tex

The routine \m{R12Reduce1} reduces a virtual diagram by performing one R2 or R1 move, if such a move is available, and otherwise it does nothing. The routine \m{R12Reduce} finds the fixed point of \m{R12Reduce1} --- in other words, it reduces a virtual diagram to using all available R1 and R2 moves.

pdf

In[∘]:=
```
R12Reduce1[vd_VD] := Tidy@Module[{R2s, R2}, Which[
    Length[R2s = Cases[vd, X_s_[i_, j_] :> X_-s[i + 1, j + 1]] ∩ (List @@ vd)] > 0,
    Complement[vd, VD[R2 = First@R2s, R2 /. X_s_[i_, j_] :> X_-s[i - 1, j - 1]]],
    Length[R2s = Cases[vd, X_s_[i_, j_] :> X_-s[i + 1, j - 1]] ∩ (List @@ vd)] > 0,
    Complement[vd, VD[R2 = First@R2s, R2 /. X_s_[i_, j_] :> X_-s[i - 1, j + 1]]],
    True, DeleteCases[vd, X_[i_, j_] /; Abs[i - j] == 1] ]];
R12Reduce[vd_VD] := FixedPoint[R12Reduce1, vd]
```

tex

Here's a very minor example:

pdf

In[∘]:=
```
VD[X_+1[1, 4], X_-1[2, 5], EOS[3], EOS[6]] // R12Reduce
```

pdf

Out[∘]= VD[EOS[1], EOS[2]]





tex

\Needspace{30mm} % 29mm is not enough.
\parpic[r]{
 \def\Xa{$\m{X}_{s_1}[i_1,j_1]$} \def\Xb{$\m{X}_{s_2}[i_2,j_2]$}
 \def\Xc{$\m{X}_{s_2}[j_1,j_2]$} \def\Xd{$\m{X}_{s_1}[i_1,i_2]$}
 \def\Xe{$\m{X}_{s_1s_2}[i_1\!-\!s_1/3,j_2\!+\!s_2/3]$} \def\Xf{$\m{X}_{-s_1s_2}[i_1\!+\!s_1/3,j_2\!-\!s_2/3]$}
 \input{figs/Gamma1.pdf_t}}
In a similar manner, \m{Γ1} performs one glide move if one is available, and \m{$\barGamma$} fully reduces under both glide moves and R1 and R2 moves. Here we bound the number of iterations by 2^{24}, to artificially stop runaway reductions such as the one in Figure~\ref{fig:swirls}.

pdf

In[∘]:=
```
Γ1[vd_VD] := Module[{js, s1, i1, j1, s2, i2, j2},
  js = Cases[vd, X_[_, j_] :> j] ⋂ Cases[vd, X_[i_, _] :> i - 1];
  If[Length[js] == 0, vd,
   j1 = RandomChoice[js]; i2 = j1 + 1;
   Cases[vd, X_s_[i_, j1] :> (s1 = s; i1 = i)];
   Cases[vd, X_s_[i2, j_] :> (s2 = s; j2 = j)];
   Tidy@Join[Complement[vd, VD[X_s1[i1, j1], X_s2[i2, j2]]],
     VD[X_s2[j1, j2], X_s1[i1, i2], X_s1 s2[i1 - s1/3, j2 + s2/3], X_-s1 s2[i1 + s1/3, j2 - s2/3]]
    ] ]]
```

In[∘]:=
```
Γ[vd_VD] := FixedPoint[Γ1, vd, 2^24]
```

In[∘]:=
```
Γ[T_] /; Head[T] =!= VD := Γ[VD[T]]
```

pdf

In[∘]:=
```
Γ̄[vd_VD] := FixedPoint[Γ1@*R12Reduce, vd, 2^24];
Γ̄[T_] /; Head[T] =!= VD := Γ̄[VD[T]]
```

tex

As expected, $\barGamma(\m{vd1})=\m{vd2}$:

pdf

In[∘]:=
```
Γ̄[vd1] == vd2
```

pdf

Out[∘]=
```
True
```

tex

Next we define the composition operation \m{d1**d2} of virtual tangle diagrams. The implementation works by ``shrinking'' \m{d2} so that each of its strands would fit between the last crossing in the corresponding strand of \m{d1} and the end of that strand, taking the union of \m{d1} and the shrank \m{d2}, and applying \m{Tidy} to the result:

pdf

In[∘]:=
```
VD /: d1_VD ** d2_VD := Tidy@Module[{es1, es2, m2},
   es1 = Cases[d1, EOS[i_] :> i];
   m2 = Max[es2 = Cases[d2, EOS[i_] :> i]];
   d1 ⋃
    Replace[DeleteCases[d2, _EOS], i_ :> i/m2 - 1 + es1[[1 + Count[es2, e_ /; i > e]]], {2}]]
```

tex

For example, ``our'' \m{vd2} has 3 crossings yet is equivalent to a 2-twist braid. So $\m{vd1}\cdot\m{vd2}$ ought to have 6 crossings while it's reduced OU form, $\barGamma(\m{vd1}\cdot\m{vd2})$ should be the Cinnamon Roll $\CR_4$, which has 7 crossings. The computer agrees:

pdf

In[∘]:= {vd2 ** vd2, $\overline{\Gamma}$[vd2 ** vd2]}

pdf

Out[∘]= {VD[EOS[9], EOS[14], X_1[1, 4], X_1[2, 11], X_1[5, 8], X_1[6, 13], X_1[10, 3], X_1[12, 7]], VD[EOS[9], EOS[16], X_1[1, 8], X_1[2, 15], X_1[3, 6], X_1[4, 13], X_1[10, 7], X_1[11, 14], X_1[12, 5]]}

Virtual Pure Braids

pdf

```
VPB[n_, {σs___}] := VPB[n, σs];
```

In[∘]:= vd

Out[∘]= VD[EOS[5], EOS[8], X_{-1}[3, 6], X_{-1}[4, 1], X_{-1}[7, 2]]

In[∘]:= vd ** vd

Out[∘]= VD[EOS[9], EOS[14], X_{-1}[3, 10], X_{-1}[4, 1], X_{-1}[7, 12], X_{-1}[8, 5], X_{-1}[11, 2], X_{-1}[13, 6]]

In[∘]:= vd ** vd ** vd

Out[∘]= VD[EOS[13], EOS[20], X_{-1}[3, 14], X_{-1}[4, 1], X_{-1}[7, 16], X_{-1}[8, 5], X_{-1}[11, 18], X_{-1}[12, 9], X_{-1}[15, 2], X_{-1}[17, 6], X_{-1}[19, 10]]

pdf

```
VD[VPB[n_]] := VD @@ (EOS /@ Range[n]);
VD[VPB[n_, σ_{i_,j_}]] := Tidy@Append[VD @@ (EOS /@ Range[n]), X_{+1}[i - 0.5, j - 0.5]];
VD[VPB[n_, σ̄_{i_,j_}]] := Tidy@Append[VD @@ (EOS /@ Range[n]), X_{-1}[i - 0.5, j - 0.5]];
VD[VPB[n_, σ_, σs__]] := VD[VPB[n, σ]] ** VD[VPB[n, σs]]
```

In[∘]:= VD[VPB[5, $\bar\sigma_{4,2}$]]

Out[∘]= VD[EOS[1], EOS[3], EOS[4], EOS[6], EOS[7], X_{-1}[5, 2]]

In[∘]:= vd1 = VD[VPB[5, $\sigma_{2,3}$]]

Out[∘]= VD[EOS[1], EOS[3], EOS[5], EOS[6], EOS[7], X_1[2, 4]]

In[∘]:= vd2 = VD[VPB[5, $\sigma_{3,4}$]]

Out[∘]= VD[EOS[1], EOS[2], EOS[4], EOS[6], EOS[7], X_1[3, 5]]

In[∘]:= VD[VPB[5, $\sigma_{2,3}$, $\sigma_{3,4}$]]

Out[∘]= VD[EOS[1], EOS[3], EOS[6], EOS[8], EOS[9], X_1[2, 4], X_1[5, 7]]

In[∘]:= VD[VPB[5, $\sigma_{2,3}$, $\sigma_{3,4}$]] // Γ

Out[∘]= VD[EOS[1], EOS[5], EOS[8], EOS[12], EOS[13], X_{-1}[4, 9], X_1[2, 11], X_1[3, 7], X_1[6, 10]]

In[∘]:= VPB[3, $\sigma_{1,2}$, $\sigma_{1,3}$, $\sigma_{2,3}$] // Γ

Out[∘]= VD[EOS[5], EOS[8], EOS[13], X_{-1}[3, 10], X_1[1, 12], X_1[2, 7], X_1[4, 9], X_1[6, 11]]

In[∘]:= VPB[3, $\sigma_{2,3}$, $\sigma_{1,3}$, $\sigma_{1,2}$] // Γ

Out[∘]= VD[EOS[3], EOS[6], EOS[9], X_1[1, 8], X_1[2, 5], X_1[4, 7]]

```
In[∘]:= R2ReduceB[vd_VD] := Module[{R2s, R2},
          R2s = Cases[vd, X_s_[i_, j_] :→ X_-s[i + 1, j + 1]] ⋂ (List @@ vd);
          If[Length[R2s] == 0, vd,
           R2 = First@R2s;
           Tidy@Complement[vd, VD[R2, R2 /. X_s_[i_, j_] :→ X_-s[i - 1, j - 1]]]
          ]]

In[∘]:= R2ReduceC[vd_VD] := Module[{R2s, R2},
          R2s = Cases[vd, X_s_[i_, j_] :→ X_-s[i + 1, j - 1]] ⋂ (List @@ vd);
          If[Length[R2s] == 0, vd,
           R2 = First@R2s;
           Tidy@Complement[vd, VD[R2, R2 /. X_s_[i_, j_] :→ X_-s[i - 1, j + 1]]]
          ]]

In[∘]:= R2Reduce[vd_VD] := FixedPoint[R2ReduceB @* R2ReduceC, vd]

In[∘]:= R1Reduce1[vd_VD] := Tidy@DeleteCases[vd, X_[i_, j_] /; Abs[i - j] == 1]

In[∘]:= VPB[3, σ_{1,2}, σ_{1,3}, σ_{2,3}] // Γ // R2Reduce

Out[∘]= VD[EOS[3], EOS[6], EOS[9], X_1[1, 8], X_1[2, 5], X_1[4, 7]]

In[∘]:= VPB[2, σ_{1,2}, σ_{2,1}] // VD

Out[∘]= VD[EOS[3], EOS[6], X_1[1, 4], X_1[5, 2]]

In[∘]:= VPB[2, σ_{1,2}, σ_{2,1}] // VD // Γ

Out[∘]= VD[EOS[7], EOS[10], X_-1[3, 4], X_1[1, 6], X_1[2, 9], X_1[8, 5]]

In[∘]:= VPB[2, σ_{1,2}, σ_{2,1}] // VD // Γ // R2Reduce

Out[∘]= VD[EOS[7], EOS[10], X_-1[3, 4], X_1[1, 6], X_1[2, 9], X_1[8, 5]]

In[∘]:= VPB[2, σ_{1,2}, σ_{2,1}] // VD // Γ // R12Reduce

Out[∘]= VD[EOS[5], EOS[8], X_1[1, 4], X_1[2, 7], X_1[6, 3]]

In[∘]:= Test1[n_, m_] := Module[{gens, i, j, k, l},
          gens = Flatten@Table[{σ_{i,j}, σ̄_{i,j}}, {i, n}, {j, DeleteCases[Range@n, i]}];
          Table[
           {i, j, k} = ijk;
           Γ̄[VPB[n, Sequence @@ p, σ_{i,j}, σ_{i,k}, σ_{j,k}, Sequence @@ q]] ==
            Γ̄[VPB[n, Sequence @@ p, σ_{j,k}, σ_{i,k}, σ_{i,j}, Sequence @@ q]],
           {l, 0, m - 3}, {p, Tuples[gens, l]}, {q, Tuples[gens, m - 3 - l]},
           {ijk, Join @@ (Permutations /@ Subsets[Range[n], {3}])}
          ]]

In[∘]:= Test1[3, 3]

Out[∘]= {{{{True, True, True, True, True, True}}}}

In[∘]:= Timing@Union@Flatten@Test1[4, 5]

Out[∘]= {127.859, {True}}
```

In[∘]:= Timing@Union@Flatten@Test1[5, 4]

Out[∘]= {9.17188, {True}}

```
In[∘]:= Test2[n_, m_] := Module[{gens, s, r = 0, ij, ijk, ijkl, perm, i, j, k, l, tests},
  gens = Flatten@Table[{σi,j, σ̄i,j}, {i, n}, {j, DeleteCases[Range@n, i]}];
  tests = Flatten[{
     Table[{i, j} = ij; {
       T[VPB[n, Join[p, {σi,j, σ̄i,j}, q]], VPB[n, Join[p, q]]],
       T[VPB[n, Join[p, {σ̄i,j, σi,j}, q]], VPB[n, Join[p, q]]]
      },
      {s, 0, m - 2}, {t, 0, s}, {p, Tuples[gens, t]}, {q, Tuples[gens, s - t]},
      {ij, Join @@ (Permutations /@ Subsets[Range[n], {2}])}
     ],
     Table[{i, j, k} = ijk; {
       T[VPB[n, Join[p, {σi,j, σi,k, σj,k}, q]], VPB[n, Join[p, {σj,k, σi,k, σi,j}, q]]],
       T[VPB[n, Join[p, {σ̄j,i, σi,k, σj,k}, q]], VPB[n, Join[p, {σj,k, σi,k, σ̄j,i}, q]]],
       T[VPB[n, Join[p, {σi,j, σi,k, σ̄k,j}, q]], VPB[n, Join[p, {σ̄k,j, σi,k, σi,j}, q]]],
       T[VPB[n, Join[p, {σi,j, σ̄k,i, σ̄k,j}, q]], VPB[n, Join[p, {σ̄k,j, σ̄k,i, σi,j}, q]]],
       T[VPB[n, Join[p, {σ̄j,i, σ̄k,i, σj,k}, q]], VPB[n, Join[p, {σj,k, σ̄k,i, σ̄j,i}, q]]],
       T[VPB[n, Join[p, {σ̄j,i, σ̄k,i, σ̄k,j}, q]], VPB[n, Join[p, {σ̄k,j, σ̄k,i, σ̄j,i}, q]]]
      },
      {s, 0, m - 3}, {t, 0, s}, {p, Tuples[gens, t]}, {q, Tuples[gens, s - t]},
      {ijk, Join @@ (Permutations /@ Subsets[Range[n], {3}])}
     ],
     Table[{i, j, k, l} = ijkl〚perm〛; {
       T[VPB[n, Join[p, {σi,j, σk,l}, q]], VPB[n, Join[p, {σk,l, σi,j}, q]]],
       T[VPB[n, Join[p, {σ̄i,j, σk,l}, q]], VPB[n, Join[p, {σk,l, σ̄i,j}, q]]],
       T[VPB[n, Join[p, {σi,j, σ̄k,l}, q]], VPB[n, Join[p, {σ̄k,l, σi,j}, q]]],
       T[VPB[n, Join[p, {σ̄i,j, σ̄k,l}, q]], VPB[n, Join[p, {σ̄k,l, σ̄i,j}, q]]]
      },
      {s, 0, m - 2}, {t, 0, s}, {p, Tuples[gens, t]}, {q, Tuples[gens, s - t]},
      {ijkl, Subsets[Range[n], {4}]}, {perm, {{1, 2, 3, 4}, {1, 3, 2, 4}, {1, 4, 2, 3}}}
     ]
    }];
  Cases[tests, T[b1_, b2_] /; Γ̄[b1] =!= Γ̄[b2]]
 ]
```

In[∘]:= Timing@Test2[3, 3]

Out[∘]= {0.34375, {}}

In[∘]:= Timing@Test2[3, 4]

Out[∘]= {7.09375, {}}

In[∘]:= Timing@Test2[3, 5]

Out[∘]= {196.203, {}}

```
In[•]:= Timing@Test2[4, 2]
Out[•]= {0.015625, {}}

In[•]:= Timing@Test2[4, 3]
Out[•]= {1.45313, {}}

In[•]:= VPB[3, σ̄1,2, σ1,3, σ2,3] // Γ // R2Reduce
Out[•]= VD[EOS[5], EOS[8], EOS[13], X-1[2, 7], X-1[3, 12], X1[1, 10], X1[4, 9], X1[6, 11]]

In[•]:= VPB[3, σ2,3, σ1,3, σ̄1,2] // Γ // R2Reduce
Out[•]= VD[EOS[3], EOS[6], EOS[9], X-1[2, 5], X1[1, 8], X1[4, 7]]

In[•]:= AllVPBInvariants[n_, m_] := Module[{gens, k},
          gens = Flatten@Table[{σi,j, σ̄i,j}, {i, n}, {j, DeleteCases[Range@n, i]}];
          Flatten@
           Table[VPB[n, Sequence @@ p] → Γ̄@VPB[n, Sequence @@ p], {k, 0, m}, {p, Tuples[gens, k]}]]

In[•]:= AllVPBInvariants[2, 2] // Column

Out[•]=
VPB[2] → VD[EOS[1], EOS[2]]
VPB[2, σ1,2] → VD[EOS[2], EOS[4], X1[1, 3]]
VPB[2, σ̄1,2] → VD[EOS[2], EOS[4], X-1[1, 3]]
VPB[2, σ2,1] → VD[EOS[2], EOS[4], X1[3, 1]]
VPB[2, σ̄2,1] → VD[EOS[2], EOS[4], X-1[3, 1]]
VPB[2, σ1,2, σ1,2] → VD[EOS[3], EOS[6], X1[1, 4], X1[2, 5]]
VPB[2, σ1,2, σ̄1,2] → VD[EOS[1], EOS[2]]
VPB[2, σ1,2, σ2,1] → VD[EOS[5], EOS[8], X1[1, 4], X1[2, 7], X1[6, 3]]
VPB[2, σ1,2, σ̄2,1] → VD[EOS[7], EOS[10], X-1[1, 4], X-1[8, 5], X1[2, 9], X1[3, 6]]
VPB[2, σ̄1,2, σ1,2] → VD[EOS[1], EOS[2]]
VPB[2, σ̄1,2, σ̄1,2] → VD[EOS[3], EOS[6], X-1[1, 4], X-1[2, 5]]
VPB[2, σ̄1,2, σ2,1] → VD[EOS[7], EOS[10], X-1[2, 9], X-1[3, 6], X1[1, 4], X1[8, 5]]
VPB[2, σ̄1,2, σ̄2,1] → VD[EOS[5], EOS[8], X-1[1, 4], X-1[2, 7], X-1[6, 3]]
VPB[2, σ2,1, σ1,2] → VD[EOS[3], EOS[8], X1[1, 6], X1[4, 7], X1[5, 2]]
VPB[2, σ2,1, σ̄1,2] → VD[EOS[3], EOS[10], X-1[1, 8], X-1[4, 7], X1[5, 2], X1[6, 9]]
VPB[2, σ2,1, σ2,1] → VD[EOS[3], EOS[6], X1[4, 1], X1[5, 2]]
VPB[2, σ2,1, σ̄2,1] → VD[EOS[1], EOS[2]]
VPB[2, σ̄2,1, σ1,2] → VD[EOS[3], EOS[10], X-1[5, 2], X-1[6, 9], X1[1, 8], X1[4, 7]]
VPB[2, σ̄2,1, σ̄1,2] → VD[EOS[3], EOS[8], X-1[1, 6], X-1[4, 7], X-1[5, 2]]
VPB[2, σ̄2,1, σ2,1] → VD[EOS[1], EOS[2]]
VPB[2, σ̄2,1, σ̄2,1] → VD[EOS[3], EOS[6], X-1[4, 1], X-1[5, 2]]
```

pdf

```
In[•]:= VPBGenerators[n_] :=
          VPBGenerators[n] = Flatten@Table[{σi,j, σ̄i,j}, {i, n}, {j, DeleteCases[Range@n, i]}];
```

In[∘]:=
```
VPBGenerators[5]
```

Out[∘]= $\{\sigma_{1,2}, \overline{\sigma}_{1,2}, \sigma_{1,3}, \overline{\sigma}_{1,3}, \sigma_{1,4}, \overline{\sigma}_{1,4}, \sigma_{1,5}, \overline{\sigma}_{1,5}, \sigma_{2,1}, \overline{\sigma}_{2,1}, \sigma_{2,3}, \overline{\sigma}_{2,3}, \sigma_{2,4}, \overline{\sigma}_{2,4}, \sigma_{2,5}, \overline{\sigma}_{2,5}, \sigma_{3,1}, \overline{\sigma}_{3,1}, \sigma_{3,2}, \overline{\sigma}_{3,2}, \sigma_{3,4}, \overline{\sigma}_{3,4}, \sigma_{3,5}, \overline{\sigma}_{3,5}, \sigma_{4,1}, \overline{\sigma}_{4,1}, \sigma_{4,2}, \overline{\sigma}_{4,2}, \sigma_{4,3}, \overline{\sigma}_{4,3}, \sigma_{4,5}, \overline{\sigma}_{4,5}, \sigma_{5,1}, \overline{\sigma}_{5,1}, \sigma_{5,2}, \overline{\sigma}_{5,2}, \sigma_{5,3}, \overline{\sigma}_{5,3}, \sigma_{5,4}, \overline{\sigma}_{5,4}\}$

In[∘]:=
```
(*CountOUForms[n_,m_]:=Module[{k},
  Length@Union@Flatten@Table[
      R12Reduce@Γ@VPB[n,Sequence@@p], {k,0,m},{p,Tuples[VPBGenerators[n],k]}]]*)
```

pdf

In[∘]:=
```
ProudFollowers[n_, σi_,j_] := ProudFollowers[n, σi,j] = Module[{p, q, s},
    Flatten@{σi,j, σj,i, σ̄j,i,
      Table[{σp,q, σq,p, σ̄p,q, σ̄q,p}, {p, {i, j}}, {q, Complement[Range[n], {i, j}]}],
      Table[{σp,q, σ̄p,q},
        {p, Complement[Range[i + 1, n], {j}]}, {q, Complement[Range[n], {i, j, p}]}]
     }];
ProudFollowers[n_, σ̄i_,j_] := ProudFollowers[n, σ̄i,j] = ProudFollowers[n, σi,j] /. σi,j → σ̄i,j
```

In[∘]:=
```
ProudFollowers[5, σ2,3]
```

Out[∘]= $\{\sigma_{2,3}, \sigma_{3,2}, \overline{\sigma}_{3,2}, \sigma_{2,1}, \sigma_{1,2}, \overline{\sigma}_{2,1}, \overline{\sigma}_{1,2}, \sigma_{2,4}, \sigma_{4,2}, \overline{\sigma}_{2,4}, \overline{\sigma}_{4,2}, \sigma_{2,5}, \sigma_{5,2}, \overline{\sigma}_{2,5}, \overline{\sigma}_{5,2}, \sigma_{3,1}, \sigma_{1,3}, \overline{\sigma}_{3,1}, \overline{\sigma}_{1,3}, \sigma_{3,4}, \sigma_{4,3}, \overline{\sigma}_{3,4}, \overline{\sigma}_{4,3}, \sigma_{3,5}, \sigma_{5,3}, \overline{\sigma}_{3,5}, \overline{\sigma}_{5,3}, \sigma_{4,1}, \overline{\sigma}_{4,1}, \sigma_{4,5}, \overline{\sigma}_{4,5}, \sigma_{5,1}, \overline{\sigma}_{5,1}, \sigma_{5,4}, \overline{\sigma}_{5,4}\}$

In[∘]:=
```
ProudFollowers[5, σ̄2,3]
```

Out[∘]= $\{\overline{\sigma}_{2,3}, \sigma_{3,2}, \overline{\sigma}_{3,2}, \sigma_{2,1}, \sigma_{1,2}, \overline{\sigma}_{2,1}, \overline{\sigma}_{1,2}, \sigma_{2,4}, \sigma_{4,2}, \overline{\sigma}_{2,4}, \overline{\sigma}_{4,2}, \sigma_{2,5}, \sigma_{5,2}, \overline{\sigma}_{2,5}, \overline{\sigma}_{5,2}, \sigma_{3,1}, \sigma_{1,3}, \overline{\sigma}_{3,1}, \overline{\sigma}_{1,3}, \sigma_{3,4}, \sigma_{4,3}, \overline{\sigma}_{3,4}, \overline{\sigma}_{4,3}, \sigma_{3,5}, \sigma_{5,3}, \overline{\sigma}_{3,5}, \overline{\sigma}_{5,3}, \sigma_{4,1}, \overline{\sigma}_{4,1}, \sigma_{4,5}, \overline{\sigma}_{4,5}, \sigma_{5,1}, \overline{\sigma}_{5,1}, \sigma_{5,4}, \overline{\sigma}_{5,4}\}$

pdf

In[∘]:=
```
ProudVPBs[n_, 0] := {VPB[n]};
ProudVPBs[n_, 1] := VPB[n, #] & /@ VPBGenerators[n];
ProudVPBs[n_, m_] /; m > 1 := Flatten[
  ProudVPBs[n, m - 1] /. VPB[n, σs___, σ_] ⧴ (VPB[n, σs, σ, #] & /@ ProudFollowers[n, σ])]
```

In[∘]:=
```
ProudVPBs[2, 2]
```

Out[∘]= $\{\text{VPB}[2, \sigma_{1,2}, \sigma_{1,2}], \text{VPB}[2, \sigma_{1,2}, \sigma_{2,1}], \text{VPB}[2, \sigma_{1,2}, \overline{\sigma}_{2,1}], \text{VPB}[2, \overline{\sigma}_{1,2}, \overline{\sigma}_{1,2}], \text{VPB}[2, \overline{\sigma}_{1,2}, \sigma_{2,1}], \text{VPB}[2, \overline{\sigma}_{1,2}, \overline{\sigma}_{2,1}], \text{VPB}[2, \sigma_{2,1}, \sigma_{2,1}], \text{VPB}[2, \sigma_{2,1}, \sigma_{1,2}], \text{VPB}[2, \sigma_{2,1}, \overline{\sigma}_{1,2}], \text{VPB}[2, \overline{\sigma}_{2,1}, \overline{\sigma}_{2,1}], \text{VPB}[2, \overline{\sigma}_{2,1}, \sigma_{1,2}], \text{VPB}[2, \overline{\sigma}_{2,1}, \overline{\sigma}_{1,2}]\}$

In[∘]:= ProudVPBs[3, 3]

Out[∘]= $\{VPB[3, \sigma_{1,2}, \sigma_{1,2}, \sigma_{1,2}], VPB[3, \sigma_{1,2}, \sigma_{1,2}, \sigma_{2,1}], VPB[3, \sigma_{1,2}, \sigma_{1,2}, \overline{\sigma}_{2,1}],$
$VPB[3, \sigma_{1,2}, \sigma_{1,2}, \sigma_{1,3}], VPB[3, \sigma_{1,2}, \sigma_{1,2}, \sigma_{3,1}], VPB[3, \sigma_{1,2}, \sigma_{1,2}, \overline{\sigma}_{1,3}],$
$VPB[3, \sigma_{1,2}, \sigma_{1,2}, \overline{\sigma}_{3,1}],$ ⋯ 1438 ⋯ $, VPB[3, \overline{\sigma}_{3,2}, \overline{\sigma}_{1,2}, \sigma_{3,1}],$
$VPB[3, \overline{\sigma}_{3,2}, \overline{\sigma}_{1,2}, \overline{\sigma}_{1,3}], VPB[3, \overline{\sigma}_{3,2}, \overline{\sigma}_{1,2}, \overline{\sigma}_{3,1}], VPB[3, \overline{\sigma}_{3,2}, \overline{\sigma}_{1,2}, \sigma_{2,3}],$
$VPB[3, \overline{\sigma}_{3,2}, \overline{\sigma}_{1,2}, \sigma_{3,2}], VPB[3, \overline{\sigma}_{3,2}, \overline{\sigma}_{1,2}, \overline{\sigma}_{2,3}], VPB[3, \overline{\sigma}_{3,2}, \overline{\sigma}_{1,2}, \overline{\sigma}_{3,2}]\}$

large output | show less | show more | show all | set size limit...

pdf

In[∘]:=
```
CountOUForms[n_, m_] := Module[{k},
  Length@Union@Flatten@Table[Γ̄@vpb, {k, 0, m}, {vpb, ProudVPBs[n, k]}]]
```

In[∘]:= Timing@CountOUForms[2, 1]

Out[∘]= {0., 5}

In[∘]:= Timing@CountOUForms[2, 2]

Out[∘]= {0., 17}

In[∘]:= Timing@CountOUForms[2, 3]

Out[∘]= {0.0625, 53}

In[∘]:= Timing@CountOUForms[2, 4]

Out[∘]= {0.28125, 161}

In[∘]:= Timing@CountOUForms[2, 5]

Out[∘]= {2.45313, 485}

In[∘]:= Timing@CountOUForms[2, 6]

Out[∘]= {25.1406, 1457}

In[∘]:= FindSequenceFunction[{5, 17, 53, 161, 485, 1457}]

Out[∘]= $-1 + 2 \times 3^{\#1}$ &

In[∘]:= FindLinearRecurrence[{5, 17, 53, 161, 485, 1457}]

Out[∘]= {4, -3}

In[∘]:= Timing@CountOUForms[3, 1]

Out[∘]= {0., 13}

In[∘]:= Timing@CountOUForms[3, 2]

Out[∘]= {0.046875, 145}

```
In[•]:= Timing@CountOUForms[3, 3]
Out[•]= {1.03125, 1561}

In[•]:= Timing@CountOUForms[3, 4]
Out[•]= {22.7813, 16 717}

In[•]:= Timing@CountOUForms[3, 5]
Out[•]= {533.859, 178 873}

In[•]:= Timing@CountOUForms[3, 6]
Out[•]= {14 058.3, 1 913 737}

In[•]:= 17038.5`^2 / 484.328125`
Out[•]= 599 409.

In[•]:= FindSequenceFunction[{13, 145, 1561, 16 717, 178 873, 1 913 737}]
Out[•]= FindSequenceFunction[{13, 145, 1561, 16 717, 178 873, 1 913 737}]

In[•]:= Timing@CountOUForms[4, 1]
Out[•]= {0., 25}

In[•]:= Timing@CountOUForms[4, 2]
Out[•]= {0.1875, 529}

In[•]:= Timing@CountOUForms[4, 3]
Out[•]= {6.8125, 10 873}

In[•]:= Timing@CountOUForms[4, 4]
Out[•]= {261.844, 222 289}

In[•]:= Timing@CountOUForms[4, 5]
Out[•]= {10 540.5, 4 540 201}

In[•]:= 9002.375`^2 / 243.4375`
Out[•]= 332 910.

In[•]:= {25, 529, 10 873, 222 289, 4 540 201}
Out[•]= {25, 529, 10 873, 222 289, 4 540 201}

In[•]:= Timing@CountOUForms[5, 1]
Out[•]= {0., 41}

In[•]:= Timing@CountOUForms[5, 2]
Out[•]= {0.484375, 1361}
```

```
In[•]:= Timing@CountOUForms[5, 3]
Out[•]= {27.25, 43121}

In[•]:= Timing@CountOUForms[5, 4]
Out[•]= {1572.84, 1351481}

In[•]:= 1459.640625`^2 / 24.515625`
Out[•]= 86905.8

In[•]:= {41, 1361, 43121, 1351481}
Out[•]= {41, 1361, 43121, 1351481}

In[•]:= Timing@CountOUForms[6, 1]
Out[•]= {0.015625, 61}

In[•]:= Timing@CountOUForms[6, 2]
Out[•]= {0.9375, 2881}

In[•]:= Timing@CountOUForms[6, 3]
Out[•]= {141.844, 127021}

In[•]:= Timing@CountOUForms[6, 4]
Out[•]= {6921.03, 5484721}

In[•]:= FindSequenceFunction@{61, 2881, 127021, 5484721}
Out[•]= FindSequenceFunction[{61, 2881, 127021, 5484721}]

In[•]:= Timing@CountOUForms[7, 1]
Out[•]= {0.03125, 85}

In[•]:= Timing@CountOUForms[7, 2]
Out[•]= {2.67188, 5377}

In[•]:= Timing@CountOUForms[7, 3]
Out[•]= {250.484, 310633}

In[•]:= {85, 5377, 310633}
Out[•]= {85, 5377, 310633}

In[•]:= Timing@CountOUForms[8, 1]
Out[•]= {0.03125, 113}

In[•]:= Timing@CountOUForms[8, 2]
Out[•]= {4.01563, 9185}
```

In[∘]:=
```
Timing@CountOUForms[8, 3]
```
Out[∘]=
```
{492.625, 668081}
```

In[∘]:=
```
{113, 9185, 668081}
```
Out[∘]=
```
{113, 9185, 668081}
```

In[∘]:=
```
Timing@CountOUForms[9, 1]
```
Out[∘]=
```
{0.03125, 145}
```

In[∘]:=
```
Timing@CountOUForms[9, 2]
```
Out[∘]=
```
{5.73438, 14689}
```

In[∘]:=
```
Timing@CountOUForms[9, 3]
```
Out[∘]=
```
{925.922, 1307233}
```

CountOUForms[n,1]:

In[∘]:=
```
n // FindSequenceFunction@{1, 5, 13, 25, 41, 61, 85, 113, 145} // Simplify // TeXForm
```
Out[∘]//TeXForm=
```
2 n^2-2 n+1
```

CountOUForms[n,2]:

In[∘]:=
```
n // FindSequenceFunction@{1, 17, 145, 529, 1361, 2881, 5377, 9185, 14689} // Simplify //
  TeXForm
```
Out[∘]//TeXForm=
```
2 n^4+4 n^3-18 n^2+12 n+1
```

CountOUForms[n,3]:

In[∘]:=
```
n // FindSequenceFunction@{1, 53, 1561, 10873, 43121, 127021, 310633, 668081, 1307233} //
   Simplify // TeXForm
```
Out[∘]//TeXForm=
```
\frac{1}{3} \left(4 n^6+36 n^5-2 n^4-546 n^3+1066 n^2-558 n+3\right)
```

In[∘]:=
```
n // FindSequenceFunction@{1, 53, 1561, 10873, 43121, 127021, 310633, 668081, 1307233}
```
Out[∘]=
$$\frac{1}{3}\left(3-558 n+1066 n^2-546 n^3-2 n^4+36 n^5+4 n^6\right)$$

In[∘]:=
$$\frac{1}{3}\left(3-558 n+1066 n^2-546 n^3-2 n^4+36 n^5+4 n^6\right) \text{ /. } n \to 1$$
Out[∘]=
```
1
```

LaTeX Epilogue

Should eventually be incorporated above and emptied.

tex

{\red MORE\hfill Tabulate the results! \hfill MORE}

\vskip 5mm
\Needspace{90mm}

```
$n$-strand classical braids with $(\leq m)$-xing:

{\def\u#1{\underline{#1}}
\setlength{\tabcolsep}{3pt}
\small\begin{tabular}{|c|c|c|c|c|c|l|}
\hline
$m\backslash n$ &   2 &   3 &      4 &      5 &      6 &      General $n$ \\
\hline
0      &   1 &   1 &      1 &      1 &      1 &      $1$ \\
1      &   3 &   5 &      7 &      9 &      11 &     {\tiny$2n-1$} \\
2      &   5 &   17 &     33 &     53 &     77 &     {\tiny$2n^2+2n-7$} \\
3      &   7 &   47 &     131 &    259 &    439 &    {\tiny$\frac{1}{3}
\left(4n^3+18n^2-22n-63\right)\ (n>2)$} \\
4      &   9 &   115 &    469 &    1143 &   2233 &   \\
5      &   11 &  263 &    1579 &   4743 &   10603 &  \\
6      &   13 &  577 &    5121 &   18941 &  48209 &  \\
7      &   15 &  1233 &   16219 &  73817 &  213119 & \\
8      &   17 &  2589 &   50581 &  283165 & 924865 & \\
9      &   19 &  5371 &   156127 & 1074963 & 3964411 & \\
$m$      &   {\tiny$2m+1$} & {\tiny$12\cdot 2^m - 2F_{m+5} - 2m - 1$\ ?}& & & &   \\
\hline
\end{tabular}}

\vskip 5mm
\Needspace{50mm}
$n$-strand (pure) virtual braids with $(\leq m)$-xing:

{\def\u#1{\underline{#1}}
\setlength{\tabcolsep}{3pt}
\small\begin{tabular}{|c|c|c|c|c|c|l|}
\hline
$m\backslash n$ &   2 &      3 &         4 &         5 &         6 &         General $n$ \\
\hline
0      &   \u{1} &   \u{1} &      \u{1} &      \u{1} &      \u{1} &      $1$ \\
1      &   \u{5} &   \u{13} &     \u{25} &     \u{41} &     \u{61} &     {\tiny$2 n^2-2
n+1$} \\
2      &   \u{17} &  \u{145} &    \u{529} &    \u{1361} &   \u{2881} &   {\tiny$2
n^4+4 n^3-18 n^2+12 n+1$} \\
3      &   \u{53} &  \u{1561} &   \u{10873} &  \u{43121} &  \u{127021} &
{\tiny$\frac{1}{3} \left(4 n^6+36 n^5-2 n^4-546 n^3+1066 n^2-558 n+3\right)$} \\
```

```
4        &    \u{161} &    \u{16717} &       \u{222289} &      1351481 &        5484721 &        \\
5        &    \u{485} &    \u{178873} &       4540201 &        &              &              \\
6        &    \u{1457} &   1913737 &          &              &              &              \\
$m$      &    {\tiny$2\cdot 3^m-1$} & &      &              &              &              \\
\hline
\end{tabular}}
```

tex

```
}
```